Music module NMS1205/00

NMS1160/00

Service
Service
Service

41 834 A12

Service Manual

SPECIFICATIONS

MUSIC MODULE NMS1205

FM Sound Synthesizer

Features	: 9 tones or 6 melodies simultaneous
	: 5 rythms
	: Vibrato
	: AM oscillators
	: 60 preset sounds

Sound sampler

Sampling method	: 4 bit ADPCM
Sampling freq.	: 16 kHz
Sampling time	: 4 seconds at 16 kHz
Sample memory	: 256K x 1 bit

Sound processor

MSX Audio processor	: Y8950

MIDI Interface

Data transfer rate	: 31.25 kBd asynchronous communication
Current loop	: 5mA
MIDI connections	: MIDI IN/OUT/THRU
MIDI connectors	: DIN 5-pin 180°

Microphone

Internal microphone	: Electret condensor
Frequency range	: 20 Hz tot 20 kHz

Ext. microphone

Connector	: Cinch plug
Input impedance	: < 50 kOhm
Input level	: 2 mV to 800 mV p/p

Audio input

Connector	: Cinch plug
Input impedance	: 50 kOhm
Input level	: 1 V p/p

Audio outputs

Connectors	: 2 Cinch plugs
Output impedance	: 50 kOhm
Output level	: 200 mV p/p

Ext. keyboard

Connector	: 20 pole socket

Power supply

	: +12 V 30 mA typ. −12 V 16 mA typ. +5 V 200 mA typ.
Power Source	: MSX computer

Dimensions

(lxwxh)	: 185x136x32 mm

MUSIC KEYBOARD NMS1160

Number of keys	: 61
Interface plug	: 20 pole female
Cable length	: 1.2 m
Dimensions (lxwxh)	: 898x185x69 mm

Safety regulations require that the set be restored to its original condition and that parts which are identical with those specified be used.

Documentation Technique Service Dokumentation Documentazione di Servizio Huolte-Ohje Manual de Servicio Manual de Serviçic

Subject to modification

4822 727 15943

Printed in The Netherlands

Published by
Service Consumer Electronics

EXPLODED VIEW



42 051 C12

MECHANICAL PARTS LIST

1	4822 432 10604	Top case
2	4822 267 30821	OUT1 CINCH conn.
2	4822 267 30821	OUT2 CINCH conn.
3	4822 256 30523	MIDI OUT DIN 5P/1800 PCB
3	4822 256 30523	MIDI IN DIN 5P/180° PCB
3	4822 256 30523	MIDI THRU DIN 5P/180° PCB
4	4822 432 10605	Bottom case
6	4822 267 30819	EXTM switched CINCH conn.
6	4822 267 30819	LINE switched CINCH conn.
7	4822 100 11159	VOL potmeter 10K
8	4822 290 60668	KEYB connector 20P

BLOCK DIAGRAM



ATTENTION

The exchange of cartridges should take place with the set turned off.

ESD

All ICs and many other semi-conductors are susceptible to electrostatic discharges (ESD). Careless handling during repair can reduce life drastically.
When repairing, make sure that you are connected with the same potential as the mass of the set via a wrist wrap with resistance.
Keep components and tools also at this potential.

INTRODUCTION

The Philips Music Module is a self-contained musical add-on to MSX and MSX-2 computer. It provides the possibillity to generate, process or sample almost any sound. The Music Module also provides a Musical Instrument Digital Interface (MIDI) which makes it possible to connect the module to other electronic musical devices provided that they are equipped with such an interface too.

TECHNICAL DESCRIPTION

The heart of the Music Module is the Yamaha Y8950 MSX Audio chip. This chip is both capable of signal sampling and of generating up to 9 simultaneous channels of frequency modulated audio sounds. The MSX Audio chip also has ports for keyboard-scanning. A line to the Z80 INT is provided.
The (digital) signal output is being fed to a Digital to Analog converter, the YM3014, which converts 10-bit serial information to an audio signal. This DA-convertor is also used in the Analog to Digital conversion process.
The sample information is being stored in an external 256k x 1 memory-chip, the TMS4256 dynamic RAM.
An additional DA-convertor, the DAC0800, is included to enable real time sound effects to be done.
The MIDI-interface consists of the MC6850-ic, an Asynchronous Communications Interface Adapter (ACIA) from the 6800-series.
All of the system software including the autostart routines is included on a 32K byte EPROM (27256).
For more information see the block diagram and the circuit diagram.

CIRCUIT DESCRIPTION

Input circuit

The analog input circuitry is built around 3 out of the 4 OPAMPS contained within IC18 (TL074). The first part of the circuit consists of the electret microphone's DC power supply; the signal then feeding into a variable gain (x3..x100) non inverting preamplifier. An external microhone facility is provided by the use of switched input socket 1. The signal is now further amplified by a factor of 6 before entering the anti aliasing filter, implemented around an OPAMP with a second order Butterworth response with a -3 dB point of 6 kHz. The resistor network R8, R29, VR1 provides the DC bias. Diodes D5, D6 provide overload protection to prevent damage to the MSX Audio chip.

Digital to Analog converters

IC14 is a 13 bit DA-converter. It has a serial input. It is connected to the MSX Audio chip with 3 lines: a serial link, a clock line and a synchronising line. The converted output is fed to a unity gain buffer (IC19-LM342) and back to the MSX Audio chip for Analog to Digital conversion purposes. Pin 7 of IC14 outputs a reference voltage of 1/2 V_{dd} which is buffered by IC19 and is used as a DC bias for various points in the output circuit.
IC15 is a 8 bit DA-converter used when the circuit is used in the ECHO and PITH SHIFTER modes. This DA converter is of the DAC0800 type and is buffered by IC1 for inputs and by IC19 for outputs.

Output circuits

The outputs of both DA converters enter analog switches. They are controlled by the MSX Audio chip multi-purpose I/0 ports and are thus under software control. IC19 provides a 16 kHz second order Butterworth response to smooth out higher harmonics. IC18 acts to mix both outputs before they enter the anti thumb circuit which delays the connection of the signal for about 1 second after power up. This is to avoid the 2.5 V bias level to appear immediately on the output causing a click. The output is also fed into the MSX computer in order to make the sound audible on a TV or Video monitor.

MIDI circuit

The MIDI interface is of an standard type (5 mA serial link operation at 31.5 kBaud, see diagram for transmission protocol) using the 6850 Asynchronous Communications Interface Adapter (ACIA) chip. The Opto Isolator used is type PC900. The ACIA clock circuit utilizes a 2 MHZ crystal in the feedback loop of two 'linearized' TTL inverters. A divide by 4 circuit constructed of 2 D-type flipflops is used to reduce this to the correct ACIA operating frequency. As the ACIA used is of the 6800-series the required enable control signal has to be generated by ANDing the Z80 Read and Write control lines. The Z80 INT is connected to the ACIA.

Address decoding

The address decoding used is standard, being performed by IC6, IC9, IC11, IC13 and IC16. A block diagram of the decoding, completed with port decoding used is included in the manual.
Note that IC16 has open collector outputs to be used to drive the MSX BUSDIR signal. The address lines are buffered by IC5, in conformity with the MSX specifications.

PCB-LAYOUT





CIRCUIT DIAGRAM

ELECTRICAL PARTS LIST MUSIC MODULE

IC1	4822 209 51254	UPD27256D-25 V1.2
IC2	5322 209 81625	74LS04N
IC3	5322 209 81623	74LS00N
IC4	4822 209 71624	68B50P
IC5	5322 209 86017	HD74LS244
IC6	5322 209 81634	HD74LS32
IC8	4822 209 51255	4256-15
IC9	5322 209 81625	74LS04N
IC10	4822 209 71622	Y8950
IC11	4822 209 83428	HD74LS30
IC12	5322 209 81646	HD74LS374
IC13	5322 209 81634	HD74LS32
IC14	4822 209 71623	Y3014
IC15	5322 209 11254	DAC08/E
IC16	5322 209 85703	74LS01N
IC17	5322 209 10357	HEF4066BP
IC18	5322 209 83581	TL074CN
IC19	4822 209 71621	CA324E
IC20	4822 130 10008	PC900
IC21	5322 209 81647	HD74LS74N



ZD1	4822 130 34167	BZY88C6V2
D1	4822 130 30621	1N4148
D2	4822 130 30621	1N4148
D3	4822 130 30621	1N4148
D4	4822 130 30621	1N4148
D5	4822 130 30621	1N4148
D6	4822 130 30621	1N4148

VR1	4822 100 11158	TRIMPOT 4.7K VERT
VOL	4822 100 11159	TRIMPOT 10K TAPER C

VARIOUS

X1	4822 242 71723	CRYSTAL 2.000MHz
MIC	4822 242 30144	ELECTRET MIKE

ELECTRICAL PARTS LIST MUSIC KEYBOARD

	4822 130 30621	1N4148

MUSIC KEYBOARD NMS1160





REPAIR METHOD

To determine causes of errors follow the Symptom/Cure list stated below. For test 1 to 9 reference is made to the diagnostic program.

Symptom:
Faulty start up or bad picture.

Cure:
EPROM malfunction: run test 1 (Checksum).
Check for bus errors.
Check edge connector.

Symptom:
Picture o.k. but no sound at all.

Cure:
Select test 5 (Music chip).
Select test 5-2 (Sinewave).

Symptom:
No real time sound effects, no pitch shift in sample playback mode.

Cure:
Select test 3 (DAC).
If the signal does not appear at all, select test 8 (Signals test), to determine possible addressing errors. If no errors found, IC15 is defective. Otherwise IC19 may be defective.

Symptom:
No sampling or faulty sampling.

Cure:
Select test 5 (Musicchip).
Select test 5-1 (9 notes).
If 9 notes test does not respond correctly it may indicate a defective Audio chip IC10. Proceed as under 'no sound at all'.
Select sample test.

Symptom:
MIDI Interface does not work or does not work correctly.

Cure:
Select test 2 (MIDI).
Connect MIDI interface lead between MIDI IN socket and MIDI OUT socket.
Select test 2-1 (ACIA).
If an error occurs during ACIA test, check data lines and control lines on IC4. To check for addressing errors the program includes test signals which can be generated by selecting the 'signals' test 8.
Select test 2 (MIDI).
Select test 2-2 (180 µs).

Symptom:
Keyboard does not work correctly.

Cure:
Select test 4 (Keyboard).

ADJUSTMENTS

VR1 Adjustment

Reference is made to Music Module Diagnostic Program test 7.

DIAGNOSTIC PROGRAM USERS MANUAL

Introduction

The diagnostic program provides a number of tests which makes it possible to determine causes of errors in the music module.
The program can also be used to test the Music Keyboard NMS 1160 using a Music Module.

The program requires:
1 A MSX computer system with at least 32 K RAM configuration.
2 An oscilloscope (15 MHz).
3 A keyboard testprint and/or a keyboard NMS 1160.
4 A MIDI connector lead
(a 5 pin DIN record/play-lead can be used).
5 A short-circuited cinch plug.
The tools described are easily obtainable or can be easily constructed. Tools 3 to 5 are specified in figure 1 to 3.
The listing of the Diagnostic Program in MSX-BASIC is shown in Appendix A.
Type in the lines 10 to 2350 and save the program on tape or disc.

List tester

Using the list-tester option.

To make sure that you have typed in the program correctly, lines 2500 to 5000 provide an easy way of checking.

To use it, add the (optional) lines 2500 to 5000 to the program. Type in RUN 2500 <return> to start the list-tester program.

On the screen the line numbers and the corresponding checksum of the line will appear. These have to match the checksums in the list of checksums (see appendix B). If not, check if there are errors in the line where the faulty checksum occurred.

The listtester will not test itself; Only lines up to 2500 will be checked.

Lines starting with REM statements will return 0 as a checksum regardless of the text after the statement. This text is optional.

The listtester ignores spaces in the program to allow you to format the program as you like.

Test procedure

Before starting the test procedure we suggest a visual check for short-circuits, etc. on the PCB of the Music Module.
Insert the Music Module in slot 1 or 2 of a switched off MSX computer and switch on the computer.

Symptom:
MSX computer does not power up when Music Module is plugged in.

Probably short-circuit
- check IC power lines
- check edge connector
- check data lines and address lines

Note:
Data line errors will result in faulty start up of the computer.

Symptom:
Music Module does not start up correctly;
Picture not correct.

- check EPROM IC1
- check edge connector

Before starting the Diagnostic program, switch off the computer, and check that the Music Module is well connected to slot 1 or 2.
Turn the computer back on while holding down the <ESC> key.
Load the Philips Music Module Diagnostics program and have it run by pressing the <F5> key.
The program will be showing its menu on the screen:

Philips Music Module Diagnostics

1. Checksum
2. MIDI
3. DAC
4. Keyboard
5. Music chip
6. Sample
7. VR1-Adjust
8. Signals
9. Start Module

Select one of above tests



Fig. 1
KEYBOARD TESTPRINT

42 056 B12



Fig. 2
MIDI INTERFACE LEAD

42 055 B12



Fig. 3
SHORT CIRCUITTED CINCH PLUG

42 054 A12

DESCRIPTION OF THE TESTS

1. Checksum

The Checksum test first will ask to which slot the Music Module is connected. Type in this information (usually 1 or 2) and hit the <RETURN>-key.
The program will return the checksum of the Music Module EPROM, which has to be DE54 for this version (UPD27256D-25 V1.2).
If not, the EPROM (IC1) is defective.

2. MIDI

The MIDI test checks the built in MIDI interface.
The test requires a MIDI Interface Lead (see fig 2).

After selecting the MIDI test the following menu will return:

1. ACIA test
2. Pulse test

The ACIA test will test automatically the MIDI Interface except IC3. (Test IC3 using the Pulse test).
First connect the MIDI OUT terminal with the MIDI IN terminal using the MIDI interface lead.
The test writes out bytes to the MIDI OUT terminal and reads them from the MIDI IN terminal at the same time.

The Pulse test will generate 180 µs pulses (see fig. 4) on the MIDI OUT terminal (IC4 pin 6) and allows to check the interface buffers using an oscilloscope.
Check these pulses on: IC4 pin 6
IC2 pin 6
MIDI OUT pin 5



180µ TEST RESPONSE ON MIDI OUT PIN 5

Fig. 4 PRS 02537 T02/727

Connect the interface lead and check the pulses on IC20 pin 2,4.
If the pulses appear on pin 2 but not on pin 4 IC20 is defective.
Check the pulses on IC3 pin 6. IC3 may be defective.

If no pulses are detected at all and data and control lines show no errors:
- Check the clock generator.
- Look for 2 MHz on IC2 pin 12. If not present, the cause may be X1 or IC2.
- Look for 1 MHz on IC21 pin 11, 5.
- Look for 0.5 MHz on IC21 pin 9. If the signal is not present, IC21 will have to be replaced.
- Look for 0.5 MHz on IC2 pin 3, 4. If the signal is not present, IC2 will have to be replaced.
- Look for pulses on IC3 pin 3. If there are none, IC3 is probably defective.

Otherwise IC4 needs replacement.

3. DAC

The DAC test allows to check the 8 bit digital to analog converter, being used for real time sound effects and for playing back samples. The test requires an oscilloscope.

Connect the oscilloscope to audio output connector OUT 1 or OUT 2. the test will generate pulses of various amplitudes. The amplitude will be shown on the screen. If no pulses are detected, the corresponding bits indicate the errors.



DAC TEST OUTPUT RESPONSE

Fig. 5 PRS 02536 T02/727

following pins should show 5 V pulses:
- IC12 pin 9 for bit 0
- IC12 pin 12 for bit 1
- IC12 pin 6 for bit 2
- IC12 pin 15 for bit 3
- IC12 pin 5 for bit 4
- IC12 pin 16 for bit 5
- IC12 pin 2 for bit 6
- IC12 pin 19 for bit 7

If these pulses appear but there is no corresponding signal on the output, test if pulses appear on:
- IC15 pin 4
- IC19 pin 14

Either of these ICs may be defective.
Check for 5 V on IC10 pin 6. If pulses appear at IC17 pin 1 but not at IC17 pin 2, IC17 must be replaced.

4. Keyboard

The keyboard test allows to test either a keyboard or the keyboard interface. It requires a keyboard (NMS1160) or a testprint. The program will ask which device is being used.

Use of the testprint will result in an automatic test of the keyboard interface. If an error occurs, the corresponding bit will be shown on the screen.
In case of error(s) check connections to the bit indicated, being both on the connector and on IC10. IC10 may be defective. Re-RUN the test to find other defects.
Use of the keyboard returns a list of numbers representing the keys pressed.
If the program does not indicate this correctly, either the keyboard or the interface (IC10, Keyboard connector) will be defective.

5. Music chip

This test is being used to check the Y8950 Sound processor (IC10). There are 2 tests:

1. 9 notes test
2. Sine wave test

The 9 notes test returns 9 subsequent guitar-like sounds. If these sounds are played correctly, correct functioning of the sound processor may be concluded.

The sine wave test generates a 1kHz sine wave which can be used to check various ICs in the output circuit. The signal on pin 21 of IC10 has a 13 bit digital form and is converted to an analog signal by IC14. The analog sine wave signal is available on IC14 pin 2.
Check for 1 kHz signal on:
- IC14 pin 2
- IC19 pin 8
- IC17 pin 8
- IC19 pin 7
- IC18 pin 7
- IC17 pin 11,3
- IC19 pin 1 should show 2.5 V
- IC10 pin 10 should show 5 V

If the signal does not appear at all:
Check for 900 kHz on IC14 pin 5
Check for 50 kHz on IC14 pin 3
If signals are not present IC10 is defective.
Otherwise check if signal on IC14 pin 4 changes when sine wave is being switched on or off.
If so, IC14 is defective.

6. Sample

This feature will test the sample function. It samples the signal on pin 4 of IC10 and allows to check the microphone, the input circuit (IC18) and the sample memory (IC8). After sampling the test will play back automatically the sample and repeat it until a key is being pressed.
The program wil ask to record a sample.
First sample through the built in microphone since this may be the cause. It is suggested, to sample a sine wave of an oscillator or a clear whistle.
If sampling through the microphone does not function correctly, try sampling through an external microphone. It may indicate errors in microphone (polarity) or diode ZD1. Look for input signal on IC18 pin 12, 14.
If signal is not O.K., check ZD1 for 6.2 V and check SKT1.

Feed a signal to the external line input. If signal on IC18 pin 1 is o.k., check IC18 pin 12, 14 and check SKT2.
If not, look for signal on IC18 pin 1,8. IC18 may be defective. VR1 may have to be recalibrated, proceed with the VR1 adjust feature.
If sampling does not work yet or sound is distorted, check the connections between IC10 en IC8. Both may be defective.

7. VR1-Adjust

This test allows to adjust the ADC (analog to digital converter) bias (VR1).
To use it connect a short-circuited cinch plug to the EXT. MIC. socket and set VOLUME to maximum.
Now turn potentiometer VR1 until the arrow displayed points to zero. Make sure that reading is stable.

8. Signals

This test generates various signals concerning the address decoding logic. It addresses the Audiochip (C0), the DAC-latch (0A) and the ACIA (00) respectively. Check for negative edge pulses on the corresponding chip select signals:

C0: pulses on	IC11 pin 8
	IC9 pin 2, 4, 6, 8, 10, 12
	IC5 pin 7, 9, 12, 14, 16, 18
0A: pulses on	IC13 pin 8, 3
	IC6 pin 11
	IC16 pin 10
00: pulses on	IC3 pin 3
	IC6 pin 6
	IC16 pin 10

Finally there is a $\overline{\text{BUSDIR}}$-test:
Check for pulses on IC16 pin 4
Either of the ICs mentioned may be defective.

Note: Finding the pulses mentioned may be uneasy because of the very short duty cycles. Using a monostable multivibrator described in this manual (Fig. 6) may help detecting the pulses.
In order to use it, connect a probe to SIGNAL IN. The signals to be tested will respond with a lighting LED.

9. Start Module

Choosing this option will result in the normal Music Module operation mode.

PULSE DETECTOR CIRCUIT



Fig. 6

DIAGNOSTIC PROGRAM LISTING

```
10 KEYOFF:SCREEN 0,0,0:COLOR 15,1,1
20 CLEAR 100,&HCFFF:DEFUSR=&HD002:GOSUB 2260
30 REM Philips music module diagnostics 3.87
40 CLS:PRINT"Philips Music Module Diagnostics"
50 PRINT"================================="
60 LOCATE 5,3:PRINT"1..Checksum
70 LOCATE 5,5:PRINT"2..MIDI
80 LOCATE 5,7:PRINT"3..DAC
90 LOCATE 5,9:PRINT"4..Keyboard
100 LOCATE 5,11:PRINT"5..Musicchip
110 LOCATE 5,13:PRINT"6..Sample
120 LOCATE 5,15:PRINT"7..VR1-adjust
130 LOCATE 5,17:PRINT"8..Signals
140 LOCATE 5,19:PRINT"9..Start Module
150 GOSUB 210:A=VAL(A$):IF A<1 OR A>9THEN 150
160 ON A GOSUB 260,330,540,660,1000,1570,1900,2090,2240
170 GOTO 40
180 REM subroutines
190 OUT &HC0,A:OUT &HC1,D:RETURN
200 OUT &HC0,A:D=INP(&HC1):RETURN
210 LOCATE 0,22:PRINT"Select one of above tests":GOTO 240
220 LOCATE 0,22:PRINT"Press any key to continue":GOTO 240
230 GOSUB 210
240 A$=INKEY$:IF A$="" THEN 240
250 RETURN
260 CLS:PRINT"Checksum ROM-test"
270 LOCATE 5,5:PRINT"To which slot is the musicmodule"
280 LOCATE 5,7:INPUT"connected";A
290 IF A<1 OR A>2 THEN 10
300 POKE &HD00E,A:A=USR(0)
310 LOCATE 5,9:PRINT"Checksum : ";HEX$(PEEK(&HD000)+256*PEEK(&HD001))
320 GOTO 220
330 CLS:PRINT"MIDI-tests"
340 LOCATE 5,4:PRINT"1..ACIA-test"
350 LOCATE 5,6:PRINT"2..Pulse-test"
360 GOSUB 210
370 A=VAL(A$):IF A<1 OR A>2 THEN 360
380 ON A GOTO 450,390
390 REM pulsetest
400 CLS:LOCATE 5,5:PRINT"180 uS pulse available"
410 LOCATE 0,22:PRINT"Press any key to abort
420 OUT(0),3:OUT(0),21
430 OUT(1),2:IF INKEY$="" THEN 430
440 RETURN
450 REM aciatest
460 CLS:LOCATE 5,5:PRINT"Make sure testconnections are Ok":GOSUB 220
470 LOCATE 5,7:PRINT"Testing ACIA"
480 OUT(0),3:OUT(0),21
490 FOR N=0 TO 255:OUT(1),N
500 RD=INP(5)
510 IF RD<>N THEN LOCATE 5,10:PRINT"ACIA ERROR":GOTO 220
520 NEXT
530 LOCATE 5,10:PRINT"ACIA ok":GOSUB220:RETURN
540 REM dactest
550 CLS:PRINT"DAC-test
560 LOCATE 5,5:PRINT"Make sure oscilloscope is
570 LOCATE 5,7:PRINT"connected to output
580 GOSUB 220
590 A=&H18:D=&HFF:GOSUB 190
600 A=&H19:D=&H1:GOSUB 190
```

```
610 FOR N=0 TO 7:D=2^N:LOCATE 5,9+N
620 PRINT"bit "N" output "D*2"mV"
630 OUT(10),D:OUT(10),0
640 IF INKEY$="" THEN 630
650 NEXT:RETURN
660 CLS:PRINT"keyboard-tests
670 LOCATE 5,5:PRINT"1..Connected to keyboard
680 LOCATE 5,7:PRINT"2..Connected to testprint
690 GOSUB 210:A=VAL(A$)
700 IF A<1 OR A>2 THEN 690
710 ON A GOTO 720,860
720 LOCATE 5,9:PRINT"Press one or more keys and
730 LOCATE 5,10:PRINT"check...
740 LOCATE 0,22:PRINT"Press any key to abort     "
750 LOCATE 4,12
760 FOR N=0 TO 7
770 OUT &HC0,6:OUT &HC1,2^N
780 OUT &HC0,5:RD=INP(&HC1)
790 RD=255-RD:IF RD=0 THEN 830
800 FOR K=0 TO 7
810 IF (RD AND 2^K)=2^K THEN PRINT N*8+K+1;:BEEP
820 NEXT
830 NEXT:PRINT STRING$(40," ")
840 IF INKEY$="" THEN 750
850 RETURN
860 LOCATE 5,9:PRINT"Make sure testprint is connected":GOSUB 220
870 DATA 01,F7,02,7F,04,FD,08,EF
880 DATA 10,BF,20,FB,40,FE,80,DF
890 RESTORE 870
900 FOR N=0 TO 7
910 READ D$:D=VAL("&H"+D$)
920 A=6:GOSUB 180:A=5:GOSUB 200
930 READ R$:R=VAL("&H"+R$)
940 IF R<>D THEN 980
950 NEXT
960 LOCATE 5,11:PRINT"Connector is Ok"
970 GOTO 220
980 LOCATE 5,11:PRINT"FAILURE OBSERVED BIT"N
990 GOTO 220
1000 REM musicchip
1010 CLS:PRINT"MSX Audio chip test"
1020 LOCATE 5,5:PRINT"1..9-notes-test"
1030 LOCATE 5,7:PRINT"2..Sinewave-test"
1040 GOSUB 210
1050 A=VAL(A$):IF A<1 OR A>2 THEN 1040
1060 ON A GOTO 1070,1400
1070 DATA 51,49,160,9,241,241,155,152,8,0
1080 RESTORE 1070:FOR N=0 TO 9:READ D(N):NEXT
1090 A=&H18:D=&HF:GOSUB 190
1100 A=&H19:D=&H8:GOSUB 190
1110 DATA 32,40,48,35,43,51,64,72,80,67,75,83,96,104,112
1120 DATA 99,107,115,128,136,144,131,139,147
1130 RESTORE 1110
1140 FOR N=0 TO 23:READ K
1150 FOR A=K TO K+2:D=D(INT(N/3)):GOSUB 190
1160 NEXT:NEXT
1170 D=D(8):FOR A=192 TO 200:GOSUB 190:NEXT
1180 D=D(9):A=189:GOSUB 190
1190 DATA 160,174,176,42
1200 DATA 161,174,177,46
```

```
1210 DATA 162,176,178,49
1220 DATA 163,2  ,179,50
1230 DATA 164,174,180,50
1240 DATA 165,174,181,54
1250 DATA 166,2  ,182,55
1260 DATA 167,174,183,57
1270 DATA 168,174,184,63
1280 DATA 176, 10,177,14
1290 DATA 178, 17,179,18
1300 DATA 180, 18,181,22
1310 DATA 182, 23,183,25
1320 DATA 184, 31
1330 RESTORE 1190
1340 LOCATE 4,10:FOR N=1 TO 9
1350 READ A,D:GOSUB 190:READ A,D:GOSUB 190
1360 PRINT N;:FOR T=1 TO 300:NEXT
1370 NEXT
1380 FOR N=1 TO 9:READ A,D:GOSUB 190:NEXT
1390 RETURN
1400 LOCATE 5,10:PRINT"Sinewave available"
1410 A=&H18:D=&HF:GOSUB 190
1420 A=&H19:D=&H8:GOSUB 190
1430 DATA 179,18,180,18,181,22,182,23,183,25,184,31
1440 DATA &H19,  8,&H20,33
1450 DATA &H23, 33,&H40,0
1460 DATA &H60,233,&H63,250
1470 DATA &H80,  0,&H83,0
1480 DATA &HC0,  1,&HA0,174
1490 DATA &HB0, 50
1500 RESTORE 1430:FOR N=1 TO 17:READ A,D:GOSUB 190:NEXT
1510 LOCATE 0,22:PRINT"Press any key to abort   "
1520 IF INKEY$="" THEN 1520
1530 RESTORE 1540
1540 DATA &h60,108,&h63,108,&H80,108,&h83,108,&hb0,18
1550 FOR N=1 TO 5:READ A,D:GOSUB 190:NEXT
1560 RETURN
1570 CLS:PRINT"Full memory sampling test"
1580 LOCATE 5,5:PRINT"Press any key to record sample
1590 IF INKEY$="" THEN 1590
1600 LOCATE 5,7:PRINT"Recording now
1610 REC=2700:RESTORE 1620
1620 DATA 19,00,04,FF,04,80,07,68
1630 DATA 08,00,09,00,0A,00,0B,FF
1640 DATA 0C,1F,0D,E1,0E,00,07,E8
1650 FOR N=1 TO 12
1660 READ A$:A=VAL("&H"+A$)
1670 READ D$:D=VAL("&H"+D$)
1680 GOSUB 190
1690 NEXT N
1700 FOR N=1 TO REC:NEXT
1710 A=7:D=&H68:GOSUB 190
1720 LOCATE 5,7:PRINT"Playing now     "
1730 RESTORE 1740
1740 DATA 19,08,04,FF,04,80,07,20
1750 DATA 08,00,09,00,0A,00,0B,FF
1760 DATA 0C,1F,10,EC,11,51,12,FF
1770 DATA 07,B0
1780 FOR N=1 TO 13
1790 READ A$:A=VAL("&H"+A$)
1800 READ D$:D=VAL("&H"+D$)
```

```
1810 GOSUB 190
1820 NEXT N
1830 FOR N=0 TO REC:NEXT
1840 LOCATE 5,7:PRINT"Repeating now    "
1850 LOCATE 0,22:PRINT"Press any key to abort    "
1860 IF INKEY$="" THEN 1860
1870 DATA 07;&HA1,07,&H20
1880 RESTORE 1870:READ A,D:GOSUB 190:READ A,D:GOSUB190
1890 RETURN
1900 CLS:PRINT"VR1-adjustment
1910 LOCATE 5,5:PRINT"Set volume to maximum and plug"
1920 LOCATE 5,7:PRINT"cinch plug in ext. mic."
1930 LOCATE 5,9:PRINT"Adjust VR1 to 0"
1940 LOCATE0,22:PRINT"Press any key to abort   "
1950 DATA &H0D,&HFF,&H0E,0,8,0,8,8,&H19,0,0
1960 RESTORE 1950
1970 FOR N=1 TO 5:READ A,D:GOSUB 190:NEXT
1980 LOCATE 5,11:PRINT"VR1 reading:"
1990 LOCATE 15,13:PRINT"0"
2000 LOCATE 5,14:PRINT"----------+----------"
2010 A=&H1A:GOSUB 200
2020 IF D<245 AND D>10 THEN 2010
2030 IF D>10 THEN X=D-256 ELSE X=D
2040 LOCATE 16+Y,15:PRINT" "
2050 LOCATE 16+X,15:PRINT"^"
2060 Y=X
2070 IF INKEY$="" THEN 2010
2080 RETURN
2090 CLS:PRINT"Signals test"
2100 O=&HC0:GOSUB 2140
2110 O=&HA:GOSUB 2140
2120 O=&H0:GOSUB 2140
2130 GOTO 2190
2140 LOCATE 5,5:PRINT"Now writing to:";
2150 PRINT HEX$(O);"H "
2160 LOCATE 0,22:PRINT"Press any key to continue"
2170 OUT(O),0:IF INKEY$="" THEN 2170
2180 RETURN
2190 LOCATE 5,5:PRINT"Check BUSDIR signal.          "
2200 LOCATE 11,4:PRINT"_____"
2210 LOCATE 0,22:PRINT"Press any key to abort.    "
2220 O=INP(C0):IF INKEY$=""THEN2220
2230 RETURN
2240 CALL MUSICBOX
2250 REM loading sumcheck-routine
2260 RESTORE 2310
2270 FOR P=&HD000 TO &HD026
2280 READ D$:POKE P,VAL("&H"+D$)
2290 NEXT
2300 RETURN
2310 DATA 00,00,F3,21,00,40,DD,21
2320 DATA 00,00,E5,DD,E5,3E,02,CD
2330 DATA 0C,00,01,00,00,DD,E1,E1
2340 DATA 4F,DD,09,23,7C,FE,C0,20
2350 DATA E9,DD,22,00,D0,FB,C9
```

LIST TESTER LISTING

```
2500 REM--------------------------
2510 REM
2520 REM  Listtester
2530 REM
2540 REM Start listtester after
2550 REM typing in the program
2560 REM  by typing "run 2500"
2570 REM    Refer to manual
2580 REM--------------------------
2590 REM
2600 CLS:PRINT"LISTTEST"
2610 PRINT:PRINT"Press any key and compare...
2620 PRINT
2630 GOSUB 5000
2640 CLS:AR=0
2650 ST=32768!
2660 PG=ST+1
2670 PN=ST+3
2680 TS=0
2685 WZ=PEEK(PG)+256*PEEK(PG+1)
2700 RG=PEEK(PN)+256*PEEK(PN+1)
2710 IF RG>2400 THEN 2900
2720 S=0
2730 FOR I=PN+2 TO WZ-1
2740 A=PEEK(I)
2750 IF A=0 THEN I=WZ
2760 IF A=143 THEN I=WZ:GOTO 2790
2770 IF A=32 THEN 2790
2780 S=(S+A) MOD 256
2790 NEXT
2800 PRINT"line: ";RG,"checksum:";S
2810 AR=AR+1
2820 IF AR= 5 THEN AR=0:GOSUB 4900
2830 TS=TS+S
2840 PG=WZ:PN=WZ+2
2850 GOTO 2685
2900 PRINT"total:";TS
2910 END
4900 PRINT:PRINT"any key to continue"
5000 IF INKEY$="" THEN 5000
5010 RETURN
```

LIST OF CHECKSUMS

LINE	SUM	LINE	SUM	LINE	SUM	LINE	SUM	LINE	SUM
10	210	20	55	30	0	40	102	50	117
60	219	70	206	80	118	90	226	100	87
110	23	120	73	130	140	140	50	150	109
160	197	170	191	180	0	190	104	200	104
210	193	220	240	230	109	240	117	250	142
260	188	270	155	280	100	290	249	300	152
310	119	320	115	330	49	340	198	350	196
360	109	370	11	380	241	390	0	400	67
410	192	420	198	430	185	440	142	450	0
460	91	470	45	480	198	490	106	500	123
510	174	520	131	530	219	540	0	550	65
560	69	570	102	580	119	590	84	600	85
610	49	620	168	630	243	640	10	650	75
660	61	670	113	680	16	690	109	700	163
710	216	720	243	730	168	740	226	750	52
760	193	770	104	780	104	790	159	800	190
810	54	820	131	830	41	840	130	850	142
860	173	870	86	880	108	890	3	900	193
910	75	920	208	930	117	940	190	950	131
960	132	970	115	980	210	990	115	1000	0
1010	252	1020	72	1030	254	1040	109	1050	182
1060	86	1070	124	1080	42	1090	84	1100	85
1110	109	1120	33	1130	244	1140	219	1150	229
1160	64	1170	223	1180	171	1190	63	1200	69
1210	76	1220	218	1230	61	1240	67	1250	220
1260	74	1270	73	1280	11	1290	26	1300	8
1310	11	1320	177	1330	68	1340	48	1350	203
1360	10	1370	131	1380	201	1390	142	1400	37
1410	84	1420	85	1430	118	1440	78	1450	67
1460	226	1470	23	1480	149	1490	245	1500	62
1510	226	1520	135	1530	164	1540	183	1550	197
1560	142	1570	245	1580	159	1590	206	1600	16
1610	169	1620	19	1630	24	1640	69	1650	197
1660	66	1670	75	1680	89	1690	209	1700	65
1710	41	1720	105	1730	108	1740	15	1750	24
1760	66	1770	137	1780	198	1790	66	1800	75
1810	89	1820	209	1830	64	1840	52	1850	226
1860	221	1870	134	1880	244	1890	142	1900	209
1910	207	1920	187	1930	154	1940	226	1950	59
1960	63	1970	197	1980	48	1990	125	2000	245
2010	86	2020	205	2030	73	2040	154	2050	247
2060	160	2070	115	2080	142	2090	63	2100	10
2110	84	2120	74	2130	45	2140	207	2150	146
2160	47	2170	199	2180	142	2190	146	2200	124
2210	16	2220	17	2230	142	2240	52	2250	0
2260	169	2270	205	2280	44	2290	131	2300	142
2310	3	2320	85	2330	32	2340	98	2350	206

total checksum: 28540